

## How to filter a dataset by one or more criteria using Excel formulas

Traditionally, filtering a dataset by multiple AND/OR conditions involved use of the Advanced Filters feature. While menu-driven, Advanced Filters can be difficult to manage and applies filters directly over the source dataset, which can be undesirable for manipulating and collaborating on the data.


A recent alternative for building complex feature is the FILTER() function, one of Excel's new dynamic arrays that create new, filtered copies of a dataset. Let's give it a try using a dataset of computer prices.

[DOWNLOAD THE EXERCISE FILE HERE](#)

You'll see that this dataset has been stored in an Excel table called *computers*. If you're not already storing your data in tables, I highly encourage you to do so. You can [learn more about getting started here](#).

To get started, here's an example of filtering by just one criterion. For example, we might want to see all the computers that are marked as premium:

```
=FILTER(computers, computers[premium] = "yes")
```

An Excel spreadsheet showing a filtered table of computer specifications. The formula bar at the top displays the formula =FILTER(computers, computers[premium] = "yes"). The table has columns labeled A through G and rows numbered 1 through 8. The data is as follows:

	A	B	C	D	E	F	G
1	price	speed	hd	ram	screen	cd	multi
2	1499	25	80	4	14	no	no
3	1795	33	85	2	14	no	no
4	1595	25	170	4	15	no	no
5	3295	33	340	16	14	no	no
6	3695	66	340	16	14	no	no
7	1720	25	170	4	14	yes	no
8	1995	50	85	2	14	no	no

To add multiple criteria to the FILTER() function, use the \* and + symbols for “and” and “or” statements, respectively.


**AND criteria.** Next, let’s look for records that are both premium *and* have a speed of greater than 30. To do so, we’ll multiply the first criterion by the second, placing each criterion in its own set of parentheses:

```
=FILTER(computers, (computers[premium]="yes") *  
(computers[speed]>30))
```

	A	B	C	D	E	F	G	H	I	J	K
A2	=FILTER(computers,(computers[premium]="yes") * (computers[speed]>30))										
1	price	speed	hd	ram	screen	cd	multi	premium	ads	trend	Index
2	1795	33	85	2	14	no	no	yes	94	1	2
3	3295	33	340	16	14	no	no	yes	94	1	5
4	3695	66	340	16	14	no	no	yes	94	1	6
5	1995	50	85	2	14	no	no	yes	94	1	8
6	2225	50	210	8	14	no	no	yes	94	1	9
7	2575	50	210	4	15	yes	no	yes	94	1	10
8	2605	66	210	8	14	no	no	yes	94	1	11

**OR criteria.** Next, we'll look for records that are either a premium computer or have a speed of at least 30. To do so, simply replace the \* with + for an OR statement:

```
=FILTER(computers, (computers[premium]="yes") +  
(computers[speed]>30))
```

An Excel spreadsheet showing the result of the FILTER function. The formula bar at the top displays the formula: =FILTER(computers,(computers[premium]="yes") + (computers[speed]>30)). The spreadsheet has columns labeled A through L and rows 1 through 8. The data is as follows:


	A	B	C	D	E	F	G	H	I	J	K	L
1	price	speed	hd	ram	screen	cd	multi	premium	ads	trend	Index	
2	1499	25	80	4	14	no	no	yes	94	1	1	
3	1795	33	85	2	14	no	no	yes	94	1	2	
4	1595	25	170	4	15	no	no	yes	94	1	3	
5	3295	33	340	16	14	no	no	yes	94	1	5	
6	3695	66	340	16	14	no	no	yes	94	1	6	
7	1720	25	170	4	14	yes	no	yes	94	1	7	
8	1995	50	85	2	14	no	no	yes	94	1	8	

**Multiple AND/OR criteria.** Just for good measure, how could we write a filter function for something like this?

*Return the records of computers that are either premium, or that have both a CD-ROM and speed of over 30.*

While this statement lists three criteria in total, we can really reduce it to two statements: the computer should be premium, or it should have both a CD-ROM and speed of over 30. For that reason, I will group the second two statements together using parentheses. I'll also use Ctrl + Enter to add carriage returns between these statements:

```
=FILTER(computers, (computers[premium]="yes") +  
((computers[cd]="yes") * (computers[speed]>30)))
```

The image shows an Excel spreadsheet with a formula bar at the top and a data table below. The formula bar contains the formula: `=FILTER(computers, (computers[premium]="yes") + ((computers[cd]="yes") * (computers[speed]>30)))`. The data table has columns labeled 'price', 'speed', 'hd', 'ram', 'screen', 'cd', 'multi', 'premium', 'ads', 'trend', and 'Index'. The rows contain numerical and categorical data for each column.

	A	B	C	D	E	F	G	H	I	J	K	L
1	price	speed	hd	ram	screen	cd	multi	premium	ads	trend	Index	
2	1499	25	80	4	14	no	no	yes	94	1	1	
3	1795	33	85	2	14	no	no	yes	94	1	2	
4	1595	25	170	4	15	no	no	yes	94	1	3	
5	3295	33	340	16	14	no	no	yes	94	1	5	
6	3695	66	340	16	14	no	no	yes	94	1	6	
7	1720	25	170	4	14	yes	no	yes	94	1	7	
8	1995	50	85	2	14	no	no	yes	94	1	8	

You can continue to add and style multiple criteria to FILTER() functions using these rules.

Pre-dynamic arrays, filtering data in Excel involved pesky menus at best and cumbersome formulas or scripts at worse. The FILTER() function combined with Excel tables provides a straightforward, scalable framework for applying multiple conditional criteria to a dataset.

## THANK YOU

Thanks for checking out this post on filtering by one or more criteria with the FILTER() function.

Do you have any questions? Let me know in the comments.

And if you're interested in getting started with dynamic array functions in Excel, check out my book, *Modern Data Analytics in*

*Excel:*

